



# Extracting Intraday Price Quotes from TAQ database in WRDS using SAS

*30 minutes interval*

(v. 2.0)

*Oscar Torres-Reyna/Todd Hines*

*otorres@princeton.edu/thines@princeton.edu*

# Note

- The code presented here is based on a sample SAS code available at WRDS (<http://wrds.wharton.upenn.edu/>)
- You can copy-and-paste the code in SAS and modify it to your needs before running.
- Make sure you copy it in the order it is presented here.
- Run it all the way from “%let wrds...” to “proc download...; run;”

# 1) Login in to WRDS from SAS

```
%let wrds = wrds.wharton.upenn.edu 4016;  
  
options comamid=TCP remote=WRDS;  
  
signon username=_prompt_;  
  
libname local 'H:\'; *This is where the data will be saved;  
  
rsubmit;  
  
libname taq'/wrds/taq/sasdata'; *Location of the data;
```

## 2) Input area

```
*%let taq_ds=taq.cq_20090102;      * data set you are interested (if only one dataset);  
  
%let start_time = '9:00:00't;  * starting time;  
  
*%let date1='02JAN2009'd;      * day you are interested (if only one);  
  
*%let interval_seconds =15*60;  * interval is 15*60 seconds, 15 min;  
  
%let interval_seconds =30*60;  * interval is 15*60 seconds, 30 min;
```

### 3) Extract selected data/date/time

The code below, extracts all available data for 2009. If you want to download specific file(s) you need to list the name of each file separately (or in weekly chunks), for example:

```
*data selected2009;  
*   set taq.cq_20090102  
*     taq.cq_20090105 - taq.cq_20090109  
*     taq.cq_20090112 - taq.cq_20090116  
*     taq.cq_20090713 - taq.cq_20090716  
*     taq.cq_20090731;
```

-----  
\*\*\*\*\* Copy code below \*\*\*\*\*;

\* Extract all data available in 2009, and 2 stocks (GE and C);

\* Time is between 9:00am and 4:30pm, retrieving SYMBOL DATE TIME and PRICE;

```
data all2009;
```

```
    set taq.cq_2009;;
```

```
    where symbol in ('GE','C') and time between '9:00:00't and '16:30:00't;
```

```
run;
```

```
proc sort; by symbol date time; run;
```

## 4) Screen data to find the trade before a set time interval

```
data xtemp2;
  set all2009;
  by symbol date time;
  format itime rtime time12.;
  if first.symbol=1 or first.date=1 then do;
    *Initialize time and price when new symbol or date starts;
    rtime=time;
    iprice=bid;
    oprice=ofr;
    itime= &start_time;
  end;
  if time >= itime then do; *Interval reached;
    output; *rtime and iprice hold the last observation values;
    itime = itime + &interval_seconds;
    do while(time >= itime); *need to fill in all time intervals;
      output;
      itime = itime + &interval_seconds;
    end;
  end;
  rtime=time;
  iprice=bid;
  oprice=ofr;
  retain itime rtime iprice oprice; *Carry time and price values forward;
  *keep symbol date itime iprice rtime;
run;
```

## 5) Check the data, save as SAS data and export to \*.csv

```
Title "Final output -- XX min interval";
proc print data=xtemp2 (obs=100);
    var symbol date itime iprice oprice rtime;
run;

*Downloading the dataset in SAS format;
*Note: change the name if you change the interval;
proc download data=WORK.Xtemp2 OUT=local.all30;
run;

*****;
* Run this separately. Exports the data to a *.csv file;

libname local 'H:\';    *This is the destination folder;

proc export data= local.all30
    outfile= 'H:\all30.csv'
    dbms=CSV REPLACE;
    putnames=YES;
run;
```